

```

*****
* PROGRAM: useOrPSWD.sas  --  Get Oracle userid and password      *
* AUTHOR: NINA WERNER  --  nina_werner@mac.com                  *
*   To use the stored encrypted password, put this in every program *
*       to call macro to retrieve the password                  *
*       and use user=&o_user  password=&o_pswd in LIBNAME          *
***** ;
%getvitals ;
;
LIBNAME  dm  ORACLE  user=&o_user  password=&o_pswd
                path=mypath      schema=myschema ;

```

```

*****
* PROGRAM: saveOrPSWD.sas  --  Save Oracle userid and password   *
* AUTHOR: NINA WERNER  --  nina_werner@mac.com                  *
* Each user must run this once to call macro to store the password *
*   You can run again when the password needs to be changed      *
*       expects userid $8 and password $8-16                    *
***** ;
*
*       ----- ;
%LET o_user  = ninal234 ;      * enter your userid ;
*
*       ----- ;
;
*
*       ----- ;
%LET o_vital = everything ;    * enter your password ;
*
*       ----- ;
%putvitals ;

```

```

*****
* PROGRAM: getOrPSWD.sas -- Decode vital information *
* AUTHOR: NINA WERNER -- nina_werner@mac.com *
* CREATED: August 2002 UNIX version *
* *
* DESCRIPTION: Required module for Oracle batch access *
* Administrator stores macro once *
* *
* DATA SET: *
* u.canopy contains vital information *
* expects userid $8 and password $8-16 *
* *
***** ;
*===== ;
* unprotect macro catalog ;
x chmod 664 "/data/sas/macro/sasmacr.sas7bcat";
*===== ;
;
%MACRO getvitals / STORE DES='NLW: retrieve Oracle user info' ;
OPTIONS NONOTES NOSOURCE NOSOURCE2 NOMACROGEN NOSYMBOLGEN NOMPRINT ;
LIBNAME u '~' ;
DATA _NULL_ ;
SET u.canopy (READ=yellow) ;
LENGTH m $64 n $32 outword $16 ;
m=INPUT(vital,$hex128.);
n=INPUT(m,$hex64.);
nlen= LENGTH(n) / 2 ;
outword = ' ' ;
DO i=1 TO nlen;
incsub = SUBSTR(n,i*2,1) ;
outword = TRIM(LEFT(incsub)) || LEFT(outword) ;
end;
CALL SYMPUT('o_user',me) ;
CALL SYMPUT('o_pswd',outword) ;
RUN ;
OPTIONS SOURCE NOTES ;
%MEND getvitals ;
;
*===== ;
* protect macro catalog after adding macro ;
x chmod 444 "/data/sas/macro/sasmacr.sas7bcat";
*===== ;

```

```

*****
* PROGRAM: putOrPSWD.sas -- Store vital information *
* AUTHOR: NINA WERNER -- nina_werner@mac.com *
* CREATED: August 2002 UNIX version *
* *
* DESCRIPTION: Required module for Oracle batch access *
* Administrator stores macro once *
* DATA SET: *
* u.canopy contains vital information *
* expects userid $8 and password $8-16 *
* *
***** ;
OPTIONS NONOTES NOSOURCE NOSOURCE2 NOMACROGEN NOSYMBOLGEN NOMPRINT ;
*===== ;
* unprotect macro catalog ;
x chmod 664 "/data/sas/macro/sasmacr.sas7bcat" ;
*===== ;
;
%MACRO putvitals / STORE DES='NLW: store Oracle user info' ;
OPTIONS NONOTES NOSOURCE NOSOURCE2 NOMACROGEN NOSYMBOLGEN NOMPRINT ;
LIBNAME u '~' ;
DATA u.canopy (KEEP=ux me vital ENCRYPT=YES
              READ=yellow ALTER=yellow)
  / PGM=store_Oracle_pswd (ALTER=noble) ;
  LENGTH incr shift $2 me $8 vstore $64 vital $128 ;
  ux="&SYSUSERID" ;
  me="&o_user" ;
  inword="&o_vital" ;
  vlen=LENGTH(inword) ;
  vstore=' ' ;
  DO i=1 TO vlen ;
    shift=SUBSTR(PUT('l%e@a&t$h!e*r#z+j%a@c&k$e!t*s#y+',
                    $HEX64.),i*2,2) ;
    incr=PUT(SUBSTR(inword,i,1),$HEX2.) ;
    vstore=LEFT(shift) || LEFT(incr) || LEFT(vstore) ;
  END ;
  vital = PUT(vstore,$HEX128.) ;
  CALL SYMPUT('vital',vital) ;
RUN ;
;

DATA PGM=work.store_Oracle_pswd ;
RUN ;
;
OPTIONS SOURCE NOTES ;
%MEND putvitals ;
;
*===== ;

```

```
* protect macro catalog after adding macro ;  
x chmod 444 "/data2/sas/shared/sasmacr.sas7bcat";  
*===== ;
```

```

*****
* PROGRAM: putOrPSWD.sas -- Store vital information *
* AUTHOR: NINA WERNER -- nina_werner@mac.com *
* CREATED: August 2002 Windows version *
*
* DESCRIPTION: Required module for Oracle batch access *
* Administrator stores macro once *
* DATA SET: *
* u.canopy contains vital information *
* expects userid $8 and password $8-16 *
*
***** ;
%MACRO putvitals / STORE DES='NLW: store Oracle user info' ;
OPTIONS NONOTES NOSOURCE NOSOURCE2 NOMACROGEN NOSYMBOLGEN NOMPRINT ;
*===== ;
%WINDOW putvital IROW=5 ROWS=19 COLUMNS=55
COLOR=WHITE
#4 @5 "Prepare for Oracle Access" COLOR=BLUE
/// @5 "Enter your Oracle userid: "
@34 o_user 8 COLOR=RED ATTR=REV_VIDEO REQUIRED=YES
// @5 "Enter your Oracle password: "
@33 o_vital 16 COLOR=RED ATTR=REV_VIDEO REQUIRED=YES
DISPLAY=NO
/// @5 "Press the ENTER Key to continue" COLOR=BLUE ;
%LET o_user = ;
%LET o_vital = ;
%DISPLAY putvital ;
;
*===== ;

LIBNAME u (sasuser) ;
DATA u.canopy (KEEP=ux me vital ENCRYPT=YES
READ=yellow ALTER=yellow)
/ PGM=store_Oracle_pswd (ALTER=noble) ;
LENGTH incr shift $2 me $8 vstore $64 vital $128 ;
ux="&SYSUSERID" ;
me="&o_user" ;
inword="&o_vital" ;
vlen=LENGTH(inword) ;
vstore=' ' ;
DO i=1 TO vlen ;
shift=SUBSTR(PUT('l%e@a&t$h!e*r#z+j%a@c&k$e!t*s#y+',
$HEX64.),i*2,2) ;
incr=PUT(SUBSTR(inword,i,1),$HEX2.) ;
vstore=LEFT(shift) || LEFT(incr) || LEFT(vstore) ;
END ;
vital = PUT(vstore,$HEX128.) ;

```

```
        CALL SYMPUT('vital',vital) ;
RUN ;
;
DATA  PGM=work.store_Oracle_pswd ;
* hide SAS dataset canopy ;
X 'ATTRIB +H "C:\Documents and Settings\&sysuserid\My
Documents\SASuser\canopy.sas7bdat"' ;
RUN ;
;
OPTIONS  SOURCE NOTES ;
%MEND putvitals ;
;
```